
Unbounded_Arrays for unbounded arrays

Package Unbounded_Arrays (see Annex) has served me long and well on many practical applications e.g. web systems (cf. my paper in Ada-Europe’2001, p. 357). It delivers what its name promises—plus a little more—in an understandable and reliable manner. The little bonus is a pair of procedures for ‘physical’ insertion and deletion. The essential items are type Unbounded_Array, procedures Assign and Resize, and functions Element and Length. Conversion to/from bounded arrays was found handy: functions To_Unbounded_Array and To_Array support this.
It is my intention to expose the package to the community, and the current workshop seems to be just the right context to do that. The way Unbounded_Arrays is designed, specified, and implemented might be of some value to the planned work there.
Unbounded arrays, also called vectors, are clearly an essential category of containers (cf. workshop proposal). Also, often they are ‘enough’. These premises seem to motivate the existence of an ‘independent’, simple package devoted to it in the library. Unbounded_Arrays might serve as a kick-off for this.
Unbounded_Arrays is loosely modelled upon Ada.Strings. This is in line with Martin Dowie’s position. On the other hand, I disagree with his proposal for making ‘non-exceptional’ versions.
What are we talking about?

I’d like to take the chance to clear up some definitions. One particular issue I’m interested in is persistency. Is it completely off the picture i.e. are containers strictly primary storage objects?
Ideas to enrich/discuss the current position, before and/or during the workshop/conference

Unbounded_Arrays planned increments. Implementation improvement. ‘Free’ code reviews at the conference. Comparison with GNAT.Table and others. Experimental and/or theoretical comparisons. With respect to performance and/or usability. Legal issues (license).
Annex. Unbounded_Arrays source-code (spec + body)
generic

  type Item_Type is private;

  Block_Size : Positive := 50;

package Unbounded_Arrays is

  type Unbounded_Array is private;

  procedure Assign

    (UA : in out Unbounded_Array;

     Index : in Positive;

     Item : in Item_Type);

  function Element

    (UA : in Unbounded_Array;

     Index : in Positive)

    return Item_Type;

  procedure Append

    (UA : in out Unbounded_Array;

     Item : in Item_Type);

  procedure Resize

    (UA : in out Unbounded_Array;

     New_Length : in Natural);

  procedure Delete

    (UA : in out Unbounded_Array;

     Index : in Positive);

  procedure Insert

    (UA : in out Unbounded_Array;

     Index : in Positive;

     Item : in Item_Type);

  function Length (UA : Unbounded_Array)

     return Natural;

  function Number_Of_Blocks (Length : Natural)
    return Natural;

  function Ceiling (Length : Natural)
    return Natural;

  type Array_Type is

   array (Positive range <>) of Item_Type;

  function To_Unbounded_Array (A : Array_Type)
    return Unbounded_Array;

  function To_Array (U : Unbounded_Array)
    return Array_Type;

private

   type Array_Ptr is access Array_Type;

   type Unbounded_Array is record

      Length : Natural := 0;

      Ptr : Array_Ptr;

   end record;

end Unbounded_Arrays;

with Unchecked_Deallocation;

package body Unbounded_Arrays is

  procedure Free is new Unchecked_Deallocation(Array_Type, Array_Ptr);

  function Number_Of_Blocks(Length: Natural) return Natural is

  begin

    return(Natural(Length - 1) / Block_Size + 1);

  exception

    when Constraint_Error =>

      return(0);

  end Number_Of_Blocks;

  function Ceiling(Length: Natural) return Natural is

  begin

    return(Number_Of_Blocks(Length) * Block_Size);

  end Ceiling;

  procedure Resize(

    UA: in out Unbounded_Array;

    New_Length: in Natural)

  is

    New_Ptr: Array_Ptr;

    N: Natural;

  begin

    if New_Length > 0 then

      New_Ptr := new Array_Type(1..Ceiling(New_Length));

      N := Natural'Min(UA.Length, New_Length);

      for I in 1..N loop

        New_Ptr.all(I) := Element(UA, I);

      end loop;

    end if;

    if UA.Ptr /= null then

      Free(UA.Ptr);

    end if;

    UA.Ptr := New_Ptr;

    UA.Length := New_Length;

  end Resize;

  procedure Assign(

    UA: in out Unbounded_Array;

    Index: in Positive;

    Item: in Item_Type) is

  begin

    if Index > Ceiling(UA.Length) then

      Resize(UA, Index);

    end if;

    if Index > UA.Length then

      UA.Length := Index;

    end if;

    UA.Ptr.all(Index) := Item;

  end Assign;

  procedure Delete(

    UA: in out Unbounded_Array;

    Index: in Positive)

  is

    New_Length: Natural := UA.Length - 1;

  begin

    if Index > UA.Length then

      raise Constraint_Error;

    end if;

    for I in Index .. New_Length loop

      Assign(UA, I, Element(UA, I + 1));

    end loop;

    if Number_Of_Blocks(New_Length) < Number_Of_Blocks(UA.Length) then

      Resize(UA, New_Length);

    end if;

    UA.Length := New_Length;

  end Delete;

  procedure Insert(

    UA: in out Unbounded_Array;

    Index: in Positive;

    Item: in Item_Type)

  is

    New_Length: Natural := UA.Length + 1;

  begin

    if Index > UA.Length then

      raise Constraint_Error;

    end if;

    if Number_Of_Blocks(New_Length) > Number_Of_Blocks(UA.Length) then

      Resize(UA, New_Length);

    end if;

    for I in reverse New_Length .. Index + 1 loop

      Assign(UA, I, Element(UA, I - 1));

    end loop;

    UA.Ptr.all(Index) := Item;

    UA.Length := New_Length;

  end Insert;

  function Element(

    UA: in Unbounded_Array;

    Index: in Positive)

    return Item_Type

  is

  begin

    if Index > UA.Length then

      raise Constraint_Error;

    end if;

    return(UA.Ptr.all(Index));

  end Element;

  function Length(UA: Unbounded_Array) return Natural is

  begin

    return(UA.Length);

  end Length;

  procedure Append(

    UA: in out Unbounded_Array;

    Item: in Item_Type) is

  begin

    Assign(UA, UA.Length + 1, Item);

  end Append;

  function To_Unbounded_Array(A: Array_Type) return Unbounded_Array is

    U: Unbounded_Array;

  begin

    for I in A'Range loop

      Append(U, A(I));

    end loop;

    return(U);

  end To_Unbounded_Array;

  function To_Array(U: Unbounded_Array) return Array_Type is

    A: Array_Type(1 .. Length(U));

  begin

    for I in A'Range loop

      A(I) := Element(U, I);

    end loop;

    return(A);

  end To_Array;

end Unbounded_Arrays;

